

# FOUNDATION FOR INTELLIGENT PHYSICAL AGENTS

## FIPA ACL Message Representation in String Specification

<b>Document title</b>	FIPA ACL Message Representation in String Specification		
<b>Document number</b>	SC000701	<b>Document source</b>	FIPA TC Agent Management
<b>Document status</b>	Standard	<b>Date of this status</b>	2002/12/03
<b>Supersedes</b>	FIPA00024		
<b>Contact</b>	fab@fipa.org		
<b>Change history</b>	See <i>Informative Annex A — ChangeLog</i>		

© 1996-2002 Foundation for Intelligent Physical Agents  
<http://www.fipa.org/>  
Geneva, Switzerland

### Notice

Use of the technologies described in this specification may infringe patents, copyrights or other intellectual property rights of FIPA Members and non-members. Nothing in this specification should be construed as granting permission to use any of the technologies described. Anyone planning to make use of technology covered by the intellectual property rights of others should first obtain permission from the holder(s) of the rights. FIPA strongly encourages anyone implementing any part of this specification to determine first whether part(s) sought to be implemented are covered by the intellectual property of others, and, if so, to obtain appropriate licenses or other permission from the holder(s) of such intellectual property prior to implementation. This specification is subject to change without notice. Neither FIPA nor any of its Members accept any responsibility whatsoever for damages or liability, direct or consequential, which may result from the use of this specification.

## 21 **Foreword**

22 The Foundation for Intelligent Physical Agents (FIPA) is an international organization that is dedicated to promoting the  
23 industry of intelligent agents by openly developing specifications supporting interoperability among agents and agent-  
24 based applications. This occurs through open collaboration among its member organizations, which are companies and  
25 universities that are active in the field of agents. FIPA makes the results of its activities available to all interested parties  
26 and intends to contribute its results to the appropriate formal standards bodies where appropriate.

27 The members of FIPA are individually and collectively committed to open competition in the development of agent-  
28 based applications, services and equipment. Membership in FIPA is open to any corporation and individual firm,  
29 partnership, governmental body or international organization without restriction. In particular, members are not bound to  
30 implement or use specific agent-based standards, recommendations and FIPA specifications by virtue of their  
31 participation in FIPA.

32 The FIPA specifications are developed through direct involvement of the FIPA membership. The status of a  
33 specification can be either Preliminary, Experimental, Standard, Deprecated or Obsolete. More detail about the process  
34 of specification may be found in the FIPA Document Policy [f-out-00000] and the FIPA Specifications Policy [f-out-  
35 00003]. A complete overview of the FIPA specifications and their current status may be found on the FIPA Web site.

36 FIPA is a non-profit association registered in Geneva, Switzerland. As of June 2002, the 56 members of FIPA  
37 represented many countries worldwide. Further information about FIPA as an organization, membership information,  
38 FIPA specifications and upcoming meetings may be found on the FIPA Web site at <http://www.fipa.org/>.

39   **Contents**

40	1	Scope.....	1
41	2	String ACL Representation .....	2
42	2.1	Component Name .....	2
43	2.2	Syntax.....	2
44	2.3	Lexical Rules .....	3
45	2.4	Representation of Time .....	4
46	2.5	Notes on the Grammar Rules.....	4
47	3	References .....	6
48	4	Informative Annex A — ChangeLog .....	7
49	4.1	2002/11/01 - version H by TC X2S.....	7
50	4.2	2002/12/03 - version I by FIPA Architecture Board .....	7

## 51 **1 Scope**

52 This document deals with message transportation between inter-operating agents and also forms part of the FIPA  
53 Agent Management Specification [FIPA00023]. It contains specifications for:

- 54
- 55 • Syntactic representation of ACL in string form.
- 56

## 2 String ACL Representation

This section defines the message transport syntax for string representation which is expressed in standard EBNF format (see *Table 1*).

Grammar rule component	Example
Terminal tokens are enclosed in double quotes	" ( "
Non-terminals are written as capitalised identifiers	Expression
Square brackets denote an optional construct	[ ", " OptionalArg ]
Vertical bars denote an alternative between choices	Integer   Float
Asterisk denotes zero or more repetitions of the preceding expression	Digit*
Plus denotes one or more repetitions of the preceding expression	Alpha+
Parentheses are used to group expansions	( A   B )*
Productions are written with the non-terminal name on the left-hand side, expansion on the right-hand side and terminated by a full stop	ANonTerminal = "terminal".

Table 1: EBNF Rules

### 2.1 Component Name

The name assigned to this component is:

fipa.acl.rep.string.std

### 2.2 Syntax

```
ACLCommunicativeAct      = Message.
Message                   = "(" MessageType
                           MessageParameter* ")" .
MessageType               = See [FIPA00037]
MessageParameter          = ":"sender" AgentIdentifier
                           | ":"receiver" AgentIdentifierSet
                           | ":"content" String
                           | ":"reply-with" Expression
                           | ":"reply-by" DateTime
                           | ":"in-reply-to" Expression
                           | ":"reply-to" AgentIdentifierSet
                           | ":"language" Expression
                           | ":"encoding" Expression
                           | ":"ontology" Expression
                           | ":"protocol" Word
                           | ":"conversation-id" Expression
                           | UserDefinedParameter Expression.
UserDefinedParameter      = Word1.
Expression                = Word
                           | String
                           | Number
                           | DateTime
                           | "(" Expression* ")" .
```

<sup>1</sup> User-defined parameters must start with "x-".

```
99 AgentIdentifier      = "(" "agent-identifier"
100                      ":name" word
101                      [ ":addresses" URLSequence ]
102                      [ ":resolvers" AgentIdentifierSequence ]
103                      ( UserDefinedParameter Expression ) * ")" .
104
105
106 AgentIdentifierSequence = "(" "sequence" AgentIdentifier* ")" .
107
108 AgentIdentifierSet      = "(" "set" AgentIdentifier* ")" .
109
110 URLSequence             = "(" "sequence" URL* ")" .
111
112 DateTime               = DateTimeToken .
113
114 URL                    = See [RFC2396]
115
```

116 **2.3 Lexical Rules**

117 Some slightly different rules apply for the generation of lexical tokens<sup>2</sup>. Lexical tokens use the same notation as above,  
118 with the exceptions noted in Table 2.  
119

Lexical rule component	Example
Square brackets enclose a character set	[ "a", "b", "c" ]
Dash in a character set denotes a range	[ "a" - "z" ]
Tilde denotes the complement of a character set if it is the first character	[ ~ "(", ")" ]
Post-fix question-mark operator denotes that the preceding lexical expression is optional (may appear zero or one times)	[ "0" - "9" ] ? [ "0" - "9" ]

```
120
121
122
123 Word = [ ~ "\0x00" - "\0x20", "(", ")", "#", "0" - "9", "-", "@" ]
124       [ ~ "\0x00" - "\0x20", "(", ")", "]" * .
125
126 String = StringLiteral | ByteLengthEncodedString .
127
128 StringLiteral = "\"\" ([ ~ "\"" ] | "\\\"")* "\"\" .
129
130 ByteLengthEncodedString = "#" Digit+ "\"\" <byte sequence> .
131
132 Number = Integer | Float .
133
134 URL = See [RFC2396]
135
136 DateTimeToken = Sign?
137                Year Month Day "T"
138                Hour Minute Second MilliSecond
139                ( TypeDesignator ? ) .
140
141 Year = Digit Digit Digit Digit .
142
143 Month = Digit Digit .
144
145 Day = Digit Digit .
146
147 Hour = Digit Digit .
148
149 Minute = Digit Digit .
```

Table 2: Lexical Rules

<sup>2</sup> All white space, tabs, carriage returns and line feeds between tokens should be skipped by the lexical analyser.

```

150
151 Second                = Digit Digit.
152
153 MilliSecond           = Digit Digit Digit.
154
155 TypeDesignator        = AlphaCharacter.
156
157 AlphaCharacter         = [ "a" - "z" ] | [ "A" - "Z" ].
158
159 Digit                  = [ "0" - "9" ].
160
161 Sign                   = [ "+" , "-" ] .
162
163 Integer                = Sign? Digit+.
164
165 Dot                    = [ "." ].
166
167 Float                  = Sign? FloatMantissa FloatExponent?
168                        | Sign? Digit+ FloatExponent
169
170 FloatMantissa          = Digit+ Dot Digit*
171                        | Digit* Dot Digit+
172
173 FloatExponent          = Exponent Sign? Digit+
174
175 Exponent                = [ "e", "E" ]
176

```

## 177 2.4 Representation of Time

178 Time tokens are based on [ISO8601], with extension for relative time and millisecond durations. Time expressions may  
 179 be absolute, or relative. Relative times are distinguished by the sign character + or – appearing as the first character in  
 180 the token. If no type designator is given, the local time zone is then used. The type designator for UTC is the character  
 181 z; UTC is preferred to prevent time zone ambiguities. Note that years must be encoded in four digits. As an example,  
 182 8:30 am on 15th April, 1996 local time would be encoded as:

```

183
184 19960415T083000000
185

```

186 The same time in UTC would be:

```

187
188 19960415T083000000Z
189

```

190 while one hour, 15 minutes and 35 milliseconds from now would be:

```

191
192 +000000000T011500035
193

```

## 194 2.5 Notes on the Grammar Rules

- 195 1. The standard definitions for integers and floating point are assumed.
- 196
- 197 2. All keywords are case-insensitive.
- 198
- 199 3. A length encoded string is a context sensitive lexical token. Its meaning is as follows: the message envelope of the  
 200 token is everything from the leading # to the separator " (inclusive). Between the markers of the message envelope  
 201 is a decimal number with at least one digit. This digit then determines that *exactly* that number of 8-bit bytes are to  
 202 be consumed as part of the token, without restriction. It is a lexical error for less than that number of bytes to be  
 203 available.
- 204

4. Note that not all implementations of the ACC (see [FIPA00067]) will support the transparent transmission of 8-bit characters. It is the responsibility of the agent to ensure, by reference to internal API of the ACC, that a given channel is able to faithfully transmit the chosen message encoding.
5. A well-formed message will obey the grammar, and in addition, will have at most one of each of the parameters. It is an error to attempt to send a message which is not well formed. Further rules on well-formed messages may be stated or implied the operational definitions of the values of parameters as these are further developed.
6. Strings encoded in accordance with [ISO2022] may contain characters which are otherwise not permitted in the definition of `Word`. These characters are ESC (0x1B), SO (0x0E) and SI (0x0F). This is due to the complexity that would result from including the full [ISO2022] grammar in the above EBNF description. Hence, despite the basic description above, a word may contain any well-formed [ISO2022] encoded character, other (representations of) parentheses, spaces, or the # character. Note that parentheses may legitimately occur as *part* of a well formed escape sequence; the preceding restriction on characters in a word refers only to the encoded characters, not the form of the encoding.
7. The format for time tokens is defined in Section 2.4.
8. The format for an AID is defined in [FIPA00023].



### 3 References

- [FIPA00023] FIPA Agent Management Specification. Foundation for Intelligent Physical Agents, 2000.  
<http://www.fipa.org/specs/fipa00023/>
- [FIPA00037] FIPA Communicative Act Library Specification. Foundation for Intelligent Physical Agents, 2000.  
<http://www.fipa.org/specs/fipa00037/>
- [FIPA00067] FIPA Agent Message Transport Service Specification. Foundation for Intelligent Physical Agents, 2000.  
<http://www.fipa.org/specs/fipa00067/>
- [FIPA00075] FIPA Agent Message Transport Protocol for IIOP Specification. Foundation for Intelligent Physical Agents, 2000.  
<http://www.fipa.org/specs/fipa00075/>
- [ISO2022] Information Technology, Character Code Structure and Extension Techniques. International Standards Organisation, 1994.  
<http://www.iso.ch/cate/d22747.html>
- [ISO8601] Date Elements and Interchange Formats, Information Interchange-Representation of Dates and Times. International Standards Organisation, 1998.  
<http://www.iso.ch/cate/d15903.html>
- [RFC2396] Uniform Resource Identifiers: Generic Syntax. Request for Comments, 1998.  
<http://www.ietf.org/rfc/rfc2396.txt>

244   **4   Informative Annex A — ChangeLog**

245   **4.1   2002/11/01 - version H by TC X2S**

246   **Page 3, line 134:**           **Fixed the definition of relative time**  
247   Page 4, line 186:           Added description of definition of relative time  
248

249   **4.2   2002/12/03 - version I by FIPA Architecture Board**

250   Entire document:           Promoted to Standard status  
251